DECUS
PROGRAM LIBRARY

DECUS NO.	8-311
TITLE	CARD TO TAPE CONVERSION WITH DIAGNOSTICS
AUTHOR	B. J. Little
COMPANY	Sandia Corporation Livermore, California
DATE	January 30, 1970
SOURCE LANGUAGE	PAL III

CARD TO TAPE CONVERSION WITH DIAGNOSTICS

DECUS Program Library Write-up

DECUS No. 8-311

ABSTRACT

THIS ROUTINE CONVERTS CARDS PUNCHED IN HOLLERITH CODE TO PAPER TAPE PUNCHED IN ASCII CODE ON EITHER HIGH OR LOW SPEED PUNCH. INPUT AND OUTPUT DEVICES ARE OPERATED AT MAXIMUM SPEED. CHARACTER VALIDITY CHECK IS MADE ON EACH CARD COLUMN. DATA LOSS CHECK IS MADE ON EACH CARD. ERROR DIAGNOSTICS ALLOW RECOVERY FROM MOST DATA LOSS, READ ERROR, AND HARDWARE FAILURE SITUATIONS. USER GENERATED CARD DECK OF KNOWN ERRORS MAY BE USED TO PERFORM OPERATIONAL CHECK OF READER.

REQUIREMENTS

A. COMPUTER--PDP-8/I WITH CR-8/I (GDI MODEL 100 CARD READER).
B. CARD DECK--CARDS TO BE CONVERTED PUNCHED IN HOLLERITH CODE.

RESTRICTIONS

ALL 64 CHARACTERS IN HOLLERITH CODE ARE RECOGNIZED AS VALID. ALPHAS, NUMERICS, AND 11 SPECIAL CHARACTERS OF IBM 26 PUNCH CHARACTER SET ARE CONVERTED TO ASCII CODE. REMAINING VALID CODES ARE CONVERTED TO ZEROES (EQUIVALENT TO BLANK TAPE).

USAGE

A. LOAD OBJECT TAPE WITH BINARY LOADER.
B. TURN ON THE DESIRED OUTPUT DEVICE. EITHER LOW OR HIGH SPEED PUNCH MAY BE USED.
C. LOAD CARD DECK IN INPUT HOPPER.
D. LOAD STARTING ADDRESS 0200 AND PRESS START SWITCH.
E. REFER TO ERROR DIAGNOSTICS IN COMMENTS SECTION IF COMPUTER HALTS BEFORE CARD TO TAPE CONVERSION IS COMPLETE.

COMMENTS

PROGRAM IS CONTAINED BETWEEN LOCATIONS 0000-0600. LOCATIONS 0600-7600 ARE USED AS BUFFER STORAGE. THE CARD READER OPERATES AT FULL SPEED FILLING THIS BUFFER. AT THE SAME TIME THE OUTPUT DEVICE IS OPERATING AT FULL SPEED EMPTYING THE BUFFER.

ALL 80 COLUMNS OF THE CARD ARE READ TWICE. FIRST IN THE BINARY MODE THEN THE ALPHA-NUMERIC MODE. THE BINARY READING IS CHECKED FOR CHARACTER VALIDITY. THE ALPHA-NUMERIC READING IS COMPARED TO THE VALID BINARY READING. A CARD COLUMN COUNT IS KEPT AND DATA LOSS IS ASSUMED IF MORE OR LESS THAN 80 COLUMNS ARE READ.

CARRIAGE-RETURN AND LINE-FEED CODES ARE GENERATED AS TERMINATORS FOR EACH CARD. THEY ARE PLACED IN THE CARD READ BUFFER IMMEDIATELY AFTER THE LAST NON-SPACE CHARACTER PUNCHED IN THE CARD. THEY ARE PLACED IN COLUMNS 1 AND 2 FOR BLANK CARDS.

OUTPUT CODE MAY BE CHANGED FROM ASCII TO ANY 64 CHARACTER CODE SET BY REPLACING THE CODE TABLE AT LOCATIONS 0400-0500 WITH THE PROPER HOLLERITH CODE TO NEW CODE CHARACTER SET.

ERROR DIAGNOSTICS

ACCUMULATOR CONTENTS	PROBABLE CAUSE OF ERROR HALT RECOVERY PROCEDURE
0000	CARD DONE FLAG--FLAG NOT SET AFTER 80 COLUMNS. RELOAD LAST CARD,PRESS CONT TO RESTART.
0100	DATA LOSS--DATA READY FLAG OCCURED FOR MORE THAN 80 COLUMNS. RELOAD LAST CARD,PRESS CONT TO RESTART.
0200	READ ERROR--INVALID CODE READ IN BINARY MODE OR BINARY AND ALPHA-NUMERIC READINGS DO NOT AGREE. RELOAD LAST CARD,PRESS CONT TO RESTART.
7661-7777	DATA LOSS--DATA READY FLAG OCCURED FOR LESS THAN 80 COLUMNS. RELOAD LAST CARD,PRESS CONT TO RESTART.
7660	THERE ARE SEVERAL CAUSES OF AN ERROR HALT WITH THIS ACCUMULATOR INDICATION. ERROR LIGHTS ON CARD READER CONTROL PANEL HELP DETERMINE CAUSE. NO LIGHTS--READ START OR MOTOR START SWITCHES NOT OPERATED. DATA READY FLAG INOPERATIVE IF CARD MOTION WAS OBSERVED. RCSE COMMAND INOPERATIVE IF CARD MOTION WAS ABSENT. ERROR LIGHTS BURNED OUT. HOPPER EMPTY--NORMAL INDICATION WHEN ALL CARDS PLACED IN INPUT HOPPER HAVE BEEN READ. IF CONVERSION IS NOT COMPLETE, LOAD CARDS IN INPUT HOPPER AND PRESS CONT. PICK FAIL--TOO MANY CARDS IN INPUT HOPPER. REMOVE SOME CARDS,PRESS CONT TO RESTART. DARK CHECK--TRANSLUCENT CARD OR READ LAMP INTENSITY TOO HIGH. LAST CARD HAD ALL VALID HOLLERITH CODES. PRESS CONT TO RESTART. DARK CHECK PLUS LIGHT CHECK--TRANSLUCENT CARD OR READ LAMP INTENSITY TOO HIGH. REMOVE CARD FROM READ STATION,RELOAD CARD,PRESS CONT TO RESTART. STACKER FAIL--CARD JAM IN TRACK BETWEEN READ STATION AND STACKER HOPPER. DETERMINE SALVAGE PROCEDURE. SYNC FAIL--THE LAST CARD WAS PROBABLY READ CORRECTLY. TRY TO CONTINUE BY PRESSING CONT. LIGHT CHECK--CARD IN READ STATION TOO LONG. MAY BE ACCOMPANIED BY INCORRECT DATA READINGS DUE TO LOSS OF BITS. IN GENERAL,LOSS OF BITS WILL STILL GENERATE VALID HOLLERITH CODES. PRESSURE ROLLERS IN READ STATION MAY BE WORN OR OUT OF ADJUSTMENT. SAFEST COURSE IS TO RELOAD DECK AND RESTART AT LOCATION 0200.

```
/PROGRAM LISTING
/CARD TO TAPE CONVERSION WITH DIAGNOSTICS
/B.J. LITTLE
/  1-30-70

/AUTOMATIC SELECTION OF HIGH OR LOW SPEED PUNCH
/HIGH SPEED PUNCH IS SELECTED IF POWER IS ON--OTHERWISE THE
/LOW SPEED PUNCH IS SELECTED

/PARAMETER ASSIGNMENTS
RCSF=6631               /SKIP ON DATA READY
RCRA=6632               /READ ALPHANUMERIC
RCRB=6634               /READ BINARY
RCSD=6671               /SKIP ON CARD DONE FLAG
RCSE=6672               /SELECT CARD READER AND SKIP IF READY
RCRD=6674               /CLEAR CARD DONE FLAG
        *10
/INDEX REGISTERS
AUTOA,  0               /BUFFER FILLING INDEX
AUTOB,  0               /BUFFER EMPTYING INDEX
AUTOC,  0               /TABLE SEARCH INDEX
        *20
/TEMPORARY REGISTERS AND FLAGS
TEM1,   0
TEM2,   0
TEMEND, 0               /TEMPORARY END OF INPUT BUFFER
COLCNT, 0               /COLUMN COUNTER
NOTRDY, 0               /NOT READY,OUT OF CARDS,OR READING ERROR
BFRFUL, 0               /BUFFER FULL
/INTERCOM REGISTERS
HARE,   HSP             /HIGH SPEED PUNCH OUTPUT
TURTLE, LSP             /TELETYPE OUTPUT
OUTPUT, 0               /OUTPUT DEVICE POINTER
BFRST,  BFRBG-1         /BUFFER START FOR AUTO INDEX
BFREND, -7456           /END OF BUFFER
ALFWD,  ALFTAB          /ASCII CODE TABLE START
PALFWD, 0               /ALPHA TABLE POINTER
POCTWD, OCTTAB+77       /OCTAL CODE TABLE END
/CONSTANTS
MNS12,  -12
MNS40,  -40
M100,   -100
M120,   -120
M240,   -240
P100,   100
CR,     215
LF,     212
/LOW SPEED PRINT AND PUNCH OUTPUT
LSP,    0
        CLA CLL
        TSF             /SKIP IF FLAG IS SET
        JMP I LSP
        TAD I AUTOB     /GET CHARACTER
        TLS             /PRINT AND PUNCH IT
        CLA CLL
        JMP I LSP
```

```
/HIGH SPEED PUNCH OUTPUT
HSP,    0
        CLA CLL
        PSF          /SKIP IF FLAG IS SET
        JMP I HSP
        TAD I AUTOB /GET CHARACTER
        PLS          /PUNCH IT
        CLA CLL
        JMP I HSP
/OUTPUT ONE CHARACTER TO SELECTED DEVICE
CHROUT, 0
        CLA CLL
        TAD AUTOB
        CMA
        TAD  TEMEND
        SZL CLA      /DATA AVAILABLE TO PRINT OR PUNCH
        JMP BYPASS   /YES
        ISZ NOTRDY   /HARDWARE NOT READY FLAG SET
        JMP .+5      /NO
        CLA CLL CMA
        DCA NOTRDY   /LEAVE THE NOT READY FLAG SET
        ISZ CHROUT   /MODIFY RETURN
        JMP I CHROUT
        DCA NOTRDY
        ISZ BFRFUL   /BUFFER FULL FLAG SET
        JMP .+3      /NO
        JMP I.+1
        BETA         /REFILL BUFFER
        DCA BFRFUL
        SKP
BYPASS, JMS I OUTPUT /PRINT OR PUNCH CHARACTER
        JMP I CHROUT
/INPUT TERMINATION ON READ ERROR
ERROR=.
        TAD COLCNT
        SNA CLA     /DATA READY AFTER 80 COLUMNS
        JMP OVR80   /YES
        DCA TEM1    /NO--DELAY TO COMPLETE CARD MOTION
        JMS CHROUT
        ISZ TEM1
        JMP .-2
        TAD P100
OVR80=.
        TAD P100
        DCA TEM2
        RCRB        /CLEAR DATA READY FLAG
        CLA CLL CMA
        DCA NOTRDY  /SET NOT READY FLAG
        TAD TEMEND
        DCA AUTOA   /REMOVE LAST CARD FROM BUFFER
        JMS CHROUT
        JMP .-1     /EMPTY THE BUFFER
        TAD TEM2    /GET ERROR INDICATOR
        HLT         /OPERATOR ACTION
        CLA CLL
        JMP I .+1
        NXTCRD      /NEXT CARD
```

```
        *200
/SELECT OUTPUT DEVICE
ALPHA=.
        KCC             /CLEAR KEYBOARD
        TLS             /SELECT PRINTER
        PLS             /SELECT PUNCH
        RCRB            /CLEAR DATA READY FLAG
        CLA CLL
        DCA TEM1
        TAD MNS12
        DCA TEM2
        PSF             /PUNCH FLAG SET
        SKP
        JMP FAST        /YES-USE HIGH SPEED PUNCH
        ISZ TEM1        /WAITED 42 MS
        JMP .-4         /NO
        ISZ TEM2        /WAITED 420 MS
        JMP .-6         /NO
        TAD TURTLE      /USE LOW SPEED OUTPUT
        SKP
FAST=.
        TAD HARE
        DCA OUTPUT      /SET POINTER TO OUTPUT DEVICE
/INITIALIZE--BUFFER IS EMPTY
BETA=.
        TAD BFRST
        DCA AUTOA       /SET READ INDEX
        TAD BFRST
        DCA AUTOB       /SET PUNCH INDEX
        TAD BFRST
        DCA TEMEND      /SET BUFFER TEMPORARY END POINTER
        DCA BFRFUL      /CLEAR BUFFER FULL FLAG
/READ NEXT CARD
NXTCRD=.
        TAD M120
        DCA COLCNT      /SET COLUMN COUNTER
        RCSE            /SKIP IF CARD READER READY
        JMP NPUTND      /RCSE FAILED
        DCA NOTRDY      /CLEAR NOT READY FLAG
/READ NEXT COLUMN
NXTCOL=.
        TAD MNS40
        DCA TEM1
        DCA TEM2
        RCSF            /SKIP ON DATA READY FLAG
        SKP             /NO DATA--WAIT FOR 2 PICK ATTEMPTS
        JMP HAVDTA      /HAVE DATA
        JMS CHROUT      /OUTPUT A CHARACTER
        ISZ TEM2        /INNER DELAY LOOP
        JMP .-5
        ISZ TEM1        /OUTER DELAY LOOP
        JMP .-7
```

```
NPUTND=.
        CLA CLL CMA
        DCA NOTRDY  /SET NOT READY FLAG
        TAD TEMEND
        DCA AUTOA   /REMOVE LAST CARD FROM BUFFER
        JMS CHROUT
        JMP .-1     /EMPTY THE BUFFER
        TAD COLCNT /GET ERROR INDICATOR
        HLT        /OPERATOR ACTION
        CLA CLL
        JMP NXTCRD  /READ IN NEXT CARD
/CHECK FOR LEGAL CHARACTER,COMPARE ALPHA AND BINARY READ
HAVDTA=.
        RCRB        /READ BINARY
        DCA TEM2    /SAVE CHARACTER
        TAD TEM2
        SMA         /MSB=4000
        RAL         /NO
        SMA CLA     /MSB=2000
        TAD MNS40   /NO--OCTAL CODE IS LESS THAN 2000
        TAD MNS40
        TAD POCTWD
        DCA AUTOC
        TAD MNS40
        DCA TEM1
HUNT=.
        TAD I AUTOC /GET OCTAL CODE
        TAD TEM2    /GET CHARACTER
        SNA CLA     /HAVE MATCH
        JMP .+4     /YES
        ISZ TEM1    /THROUGH SEARCHING
        JMP HUNT    /NO
        JMP ERROR   /ILLEGAL CODE
        TAD AUTOC   /GET OCTAL CODE ADDRESS
        TAD M100
        DCA PALFWD  /ALPHA TABLE POINTER
        JMS CHROUT
        RCRA        /READ ALPHA-NUMERIC
        TAD ALFWD
        CMA  IAC
        TAD PALFWD
        SZA CLA
        JMP ERROR   /ALPHA AND OCTAL READ DO NOT MATCH
        TAD I PALFWD /GET ASCII CODE
        DCA I AUTOA /STORE IN BUFFER
        ISZ COLCNT  /FINISHED 80 COLS
        JMP NXTCOL  /NO

/CHECK FOR CARD DONE FLAG AND MORE THAN 80 COLUMNS
        DCA TEM1
        ISZ TEM1    /DELAY FOR CARD DONE FLAG
        SKP
        JMP NPUTND  /CARD DONE FLAG MISSING
        RCSF
        SKP
        JMP ERROR   /DATA READY FLAG AFTER COL 80
        JMS CHROUT
        RCSD        /SKIP ON CARD DONE FLAG
        JMP .-10
        ISZ AUTOA   /SET UP FOR LAST(NON-SPACE) CHARACTER
```

```
/LOOK FOR LAST NON SPACE CHARACTER,STORE A CR AND LF
LSTCHR=.
        CLA CLL CMA RAL /SET ACC=-2
        TAD AUTOA
        DCA AUTOA    /BACK UP TWO
        TAD I AUTOA /GET CHARACTER
        TAD M240
        SNA CLA      /WAS IT A SPACE
        JMP LSTCHR   /YES
        TAD CR
        DCA I AUTOA /STORE CARRIAGE RETURN
        TAD LF
        DCA I AUTOA /STORE LINE FEED
        TAD AUTOA
        DCA TEMEND   /LAST CARD IS READY TO PRINT OR PUNCH
        CLA CLL
        TAD BFREND
        TAD AUTOA
        SNL CLA      /BUFFER FULL
        JMP NXTCRD   /NO
        CLA CLL CMA
        DCA BFRFUL   /SET BUFFER FULL FLAG
        JMS CHROUT
        JMP .-1      /EMPTY THE BUFFER
        HLT
```

```
/HOLLERITH TO ASCII CODE TABLE OF 64 DECIMAL CHARACTERS
/LISTED IN 00 TO 77 OCTAL NUMERICAL ORDER CORRESPONDING TO
/INTERNAL CARD READER CODE AS READ IN ALPHA-NUMERIC MODE
/NS(NON-SPECIFIED) CHARACTERS FOR LEGAL HOLLERITH CODES
/PRODUCE 0(EQUIVALENT TO BLANK TAPE) IN ASCII CODE.
        *400
ALFTAB=.              /CODE TABLE FOR ALPHA-NUMERIC MODE
/ASCII  CHAR-   HOLLERITH
/CODE   ACTER   CODE
  240   /SPACE  BLANK
  261   /1      1
  262   /2      2
  263   /3      3
  264   /4      4
  265   /5      5
  266   /6      6
  267   /7      7
  270   /8      8
  271   /9      9
  0     /NS     8 2
  275   /=      8 3
  247   /'      8 4
  0     /NS     8 5
  0     /NS     8 6
  0     /NS     8 7
  260   /0      0
  257   //      0 1
  323   /S      0 2
  324   /T      0 3
  325   /U      0 4
  326   /V      0 5
  327   /W      0 6
  330   /X      0 7
  331   /Y      0 8
  332   /Z      0 9
  0     /NS     0 8 2
  254   /.      0 8 3
  250   /(      0 8 4
  0     /NS     0 8 5
  0     /NS     0 8 6
  0     /NS     0 8 7
```

/ASCII /CODE	CHAR-ACTER	HOLLERITH CODE
255	/-	11
312	/J	11 1
313	/K	11 2
314	/L	11 3
315	/M	11 4
316	/N	11 5
317	/O	11 6
320	/P	11 7
321	/Q	11 8
322	/R	11 9
0	/NS	11 8 2
244	/$	11 8 3
252	/*	11 8 4
0	/NS	11 8 5
0	/NS	11 8 6
0	/NS	11 8 7
253	/+	12
301	/A	12 1
302	/B	12 2
303	/C	12 3
304	/D	12 4
305	/E	12 5
306	/F	12 6
307	/G	12 7
310	/H	12 8
311	/I	12 9
0	/NS	12 8 2
256	/.	12 8 3
251	/)	12 8 4
0	/NS	12 8 5
0	/NS	12 8 6
0	/NS	12 8 7

```
/HOLLERITH TO OCTAL CODE TABLE OF 64 DECIMAL CHARACTERS
/LISTED IN 00 TO 77 OCTAL NUMERICAL ORDER CORRESPONDING TO
/INTERNAL CARD READER CODE AS READ IN ALPHA-NUMERIC MODE
        *500
OCTTAB=.                /CODE TABLE FOR BINARY MODE
/OCTAL CHAR-   HOLLERITH
/CODE  ACTER   CODE
 0000  /SPACE BLANK
-0400  /1      1
-0200  /2      2
-0100  /3      3
-0040  /4      4
-0020  /5      5
-0010  /6      6
-0004  /7      7
-0002  /8      8
-0001  /9      9
-0202  /NS     8 2
-0102  /=      8 3
-0042  /'      8 4
-0022  /NS     8 5
-0012  /NS     8 6
-0006  /NS     8 7
-1000  /0      0
-1400  //      0 1
-1200  /S      0 2
-1100  /T      0 3
-1040  /U      0 4
-1020  /V      0 5
-1010  /W      0 6
-1004  /X      0 7
-1002  /Y      0 8
-1001  /Z      0 9
-1202  /NS     0 8 2
-1102  /,      0 8 3
-1042  /(      0 8 4
-1022  /NS     0 8 5
-1012  /NS     0 8 6
-1006  /NS     0 8 7
```

```
/OCTAL CHAR-  HOLLERITH
/CODE  ACTER  CODE
-2000  /-     11
-2400  /J     11 1
-2200  /K     11 2
-2100  /L     11 3
-2040  /M     11 4
-2020  /N     11 5
-2010  /O     11 6
-2004  /P     11 7
-2002  /Q     11 8
-2001  /R     11 9
-2202  /NS    11 8 2
-2102  /$     11 8 3
-2042  /*     11 8 4
-2022  /NS    11 8 5
-2012  /NS    11 8 6
-2006  /NS    11 8 7
-4000  /+     12
-4400  /A     12 1
-4200  /B     12 2
-4100  /C     12 3
-4040  /D     12 4
-4020  /E     12 5
-4010  /F     12 6
-4004  /G     12 7
-4002  /H     12 8
-4001  /I     12 9
-4202  /NS    12 8 2
-4102  /.     12 8 3
-4042  /)     12 8 4
-4022  /NS    12 8 5
-4012  /NS    12 8 6
-4006  /NS    12 8 7
BFRBG=.                 /BUFFER START
      $
```